

# SCIENCE & VIE

N° 866

NOVEMBRE 1989

## A LA RECHERCHE DES CIVILISATIONS PERDUES

● 1<sup>re</sup> victoire  
écologique :  
le plastique  
en maïs

● Un étrange  
corps céleste  
à réaction

M 2578 - 866 - 18,00 F



3792578018002 08660

130 FB - 5,20 FS - Scan 3,50 - 550 Ptas - 480 Esc - 20 Dh - 2,300 Dt - 4200 L - USA NYC \$ 3,75 - BCL - 1370 CFA - 845 CFP

# Base orbitale

INFORMATIQUE AMUSANTE

Le programme de jeu que nous vous proposons ce mois-ci nous a été inspiré par la lettre de l'un de nos lecteurs. Ce dernier, qui souhaite garder l'anonymat, se demandait ce qui se passait lorsque l'on se déplaçait dans une station orbitale où une pesanteur artificielle était créée par rotation de celle-ci. Nous avons donc simulé une telle situation. Certes, les dimensions de notre base restent re-

lativement modestes, elle ne mesure "que" 60 mètres de diamètre, mais cela nous permet de présenter, en valeur réelle, l'évolution de la gravité tout en conservant des vitesses de déplacement dans la base plausible: le simple fait de courir fait déjà évoluer de façon notable la valeur de la pesanteur artificielle.

Afin de conserver un aspect ludique à ce programme, nous avons

décidé que notre base serait une ferme de l'espace expérimentale. Deux problèmes vous seront donc proposés: l'organisation du jardin potager et le transport de poussins. Evidemment, vous vous apercevrez que dans les deux cas G ne doit jamais dépasser certaines bornes. Notons que, les calculs s'effectuant sur des valeurs réelles, nous vous conseillons vivement de répondre aux questions posées en vous aidant d'un papier et d'un crayon afin de ne pas commettre d'épouvantables bévues. Enfin, si nous ne proposons ici que deux types de problèmes, c'est simplement pour une raison de place;

```
10 CLS:MODE 1
20 SPEED INK 6,2
23 REM *****
24 REM *
25 REM * DEMANDE DE CALCULS EN DEGRES. *
26 REM *
27 REM *****
30 DEG
32 REM *****
33 REM *
34 REM * DIMENSIONNEMENT DES TABLEAUX *
35 REM * NECESSAIRES A LA MEMORISATION DES *
36 REM * DONNEES DE LA ROTATION DE LA BASE *
37 REM *
38 REM *****
40 DIM J(30)
50 DIM K(30)
60 DIM L(30)
70 DIM M(30)
80 DIM N(30)
90 DIM O(30)
100 DIM P(30)
110 DIM Q(30)
112 REM *****
113 REM *
114 REM * PAGE DE REMPLISSAGE COMBLANT LE *
115 REM * TEMPS DE CALCUL DES DONNEES *
116 REM *
117 REM *****
120 PRINT "ICI LE POSTE CENTRAL DE CONTROLE;":PRINT
T
130 PRINT "NOUS SOMMES EN PHASE D'APPROCHE":PRINT
140 PRINT "DE LA FERME INTERGALACTIQUE.":PRINT
150 PRINT "DANS QUELQUES INSTANTS LE CONTACT":PRIN
T
160 PRINT "RADAR VA ETRE ETABLI.":PRINT:PRINT:PRIN
T
170 PRINT "NOUS VOUS FOURNIRONS UNE":PRINT
180 PRINT "IMAGE DES QUE POSSIBLE."
182 REM *****
183 REM *
184 REM * CALCUL DES DONNEES DE L'ANIMATION *
185 REM * ET MEMORISATION PAR PAS DE 6 DEG. *
186 REM *
187 REM *****
190 FOR R=1 TO 30
200 LET I=R*6
210 LET J(R)=INT(200+(10*SIN(I+90))+(155*SIN(I))):
LET K(R)=INT(200+(10*COS(I+90))+(140*COS(I)))
220 LET L(R)=INT(200+(10*SIN(I+90))+(45*SIN(I))):L
ET M(R)=INT(200+(10*COS(I+90))+(41*COS(I)))
230 LET N(R)=INT(200+(10*SIN(I-90))+(155*SIN(I))):
LET O(R)=INT(200+(10*COS(I-90))+(140*COS(I)))
240 LET P(R)=INT(200+(10*SIN(I-90))+(45*SIN(I))):L
ET Q(R)=INT(200+(10*COS(I-90))+(41*COS(I)))
250 SOUND 1,20,5,15,0,0,0
260 SOUND 3,40,2,15,0,0,0
270 NEXT R
272 REM *****
273 REM *
274 REM * TRACE DES CERCLES DE LA BASE. *
275 REM *
276 REM *****
280 SOUND 1,100,100,15,0,0,0:SOUND 2,102,100,15,0,0,0:FOR T=1 TO 1000:NEXT T
290 MODE 0:PRINT "LIAISON EN COURS":LET IK=14
300 LET D=200:GOSUB 2180
310 LET D=160:GOSUB 2180
320 LET D=40:GOSUB 2180:SOUND 1,100,50,15,0,0,0:SO
UND 2,102,50,15,0,0,0
322 REM *****
323 REM *
324 REM * ANIMATION TEMPS REEL DE LA BASE. *
325 REM *
326 REM *****
330 LOCATE 1,25:PRINT "VOICI VOTRE BASE.":LOCATE 1
,1:PRINT "CONTACT RADAR ETABLI"
340 LOCATE 15,15:PRINT "TAPER"
350 LOCATE 15,17:PRINT "'P'"
360 LOCATE 15,19:PRINT "SVP."
370 FOR A=1 TO 30
380 LET Z=A:LET ID=4:GOSUB 2280
390 SOUND 1,150,20,15,0,0,15
400 SOUND 2,151,20,15,0,0,5
410 LET Z=A:LET ID=0:GOSUB 2280
412 REM *****
413 REM *
414 REM * DEPART JEU SUR FRAPPE DE TOUCHE *
415 REM *
416 REM *****
420 IF INKEY$<>" " THEN LET A=30:GOTO 450
430 NEXT A
440 GOTO 370
450 MODE 2:LET IK= 1
452 REM *****
453 REM *
454 REM * TRACE DES CERCLES *
455 REM *
456 REM *****
460 LET D=200:GOSUB 2180
470 LET D=160:GOSUB 2180
480 LET D=40:GOSUB 2180
482 REM *****
483 REM *
484 REM * RESERVATION DE FENETRE DE TEXTE. *
485 REM *
486 REM *****
490 WINDOW #1,55,80,1,25
502 REM *****
503 REM *
504 REM * TRACE DES DROITES *
505 REM *
506 REM *****
510 LET Z=15:LET ID=1:GOSUB 2280
512 REM *****
513 REM *
514 REM * AFFICHAGE CARACTERISTIQUES BASE *
515 REM *
516 REM *****
520 PRINT #1,"BIENVENUE A BORD DE LA "
530 PRINT #1,"BASE SVJ 89DX102.":PRINT #1
540 PRINT #1,"VOICI LE RAPPEL DE SES"
550 PRINT #1,"CARACTERISTIQUES"
560 PRINT #1,"PRINCIPALES:":SOUND 1,100,100,15,0,0,0
:SOUND 2,102,100,15,0,0,0:FOR T=1 TO 5000:NEXT T
:CLS #1
570 PRINT #1,"DIAMETRE TOTAL: 60 M"
580 PRINT #1,"MASSE TOTALE: 6 000 T"
```

notre programme est déjà relativement long. Cependant rien ne vous empêchera d'inventer bien d'autres types de situations lorsque vous serez familiarisés avec l'évolution de G en fonction de la position occupée dans la base et des vitesses de déplacement.

Cela dit, passons à l'écriture de notre programme. Comme on peut s'en douter il sera essentiellement constitué de texte.

Cependant, des lignes de 40 à 110, nous dimensionnerons 8 tableaux. En fait ceux-ci n'interviennent pas dans le déroulement du jeu lui-même mais permettent d'assurer l'animation en temps réel de

la rotation de la base lors de sa présentation. En effet les calculs des coordonnées de chaque tracé étant relativement longs à effectuer, les lignes de 190 à 270 se chargent de cette tâche, nous les effectuerons d'avance et, pour l'animation, l'ordinateur n'aura plus qu'à venir piocher les données dans nos tableaux.

Vient ensuite cette animation. Nous utiliserons ici deux sous-routines que nous examinerons plus tard. La ligne 420 permet de sortir de cette animation pour commencer le jeu sur la frappe d'une touche. Nous tracerons alors de nouveau les contours de notre

base, mais cette fois-ci avec une meilleure résolution, et une fenêtre sera réservée pour inscrire nos commentaires. Le texte de présentation des caractéristiques de votre station occupe les lignes de 520 à 770. Notons qu'il est très important de le lire attentivement : il contient toutes les données qui vous seront nécessaires pour résoudre les problèmes posés.

Le premier problème est présenté par les lignes de 780 à 1260. Notons que l'ensemble des ordres PRINT seront suivis de 31; cela afin de bien inscrire le texte dans la fenêtre réservée. Vient alors la prise en compte de votre réponse

```

590 PRINT #1, "DUREE REVOLUTION:10,99 s":PRINT #1
600 PRINT #1, "REPARTITION DES ESPACES":PRINT #1
610 PRINT #1, "COURSIVE PRINCIPALE:"
620 PRINT #1, "CIRCONFERENCE: 188,49 M":PRINT #1, "H
AUTEUR PLAFOND: 6M"
630 PRINT #1, "VITESSE TANGE.: 17,15 M/s":PRINT #1,
"(61,74 km/h)"
640 PRINT #1, "PESANTEUR ARTIF: 9,805":PRINT #1
650 PRINT #1, "SAS D'ACCUEIL:"
660 PRINT #1, "DIAMETRE: 12 M"
670 PRINT #1, "CIRCONFERENCE: 37,7M"
680 PRINT #1, "VITESSE TANGE.: 3,42 M/s":PRINT #1, "
(12,31 km/h)"
690 PRINT #1, "PESANTEUR ARTIF.: 1,95":PRINT #1
700 PRINT #1, "JONCTIONS DE LIAISONS"
710 PRINT #1, "LONGUEUR: 16 M"
720 PRINT #1, "DIAMETRE: 3 M"
730 PRINT #1, "PESANTEUR: 1,96 A 7,19"
740 PRINT "POUR CONTINUER TAPER UNE TOUCHE."
750 SOUND 1,100,200,15,0,0,0
760 SOUND 2,102,200,15,0,0,0
770 IF INKEY$="" THEN GOTO 770
772 REM *****
773 REM *
774 REM * PREMIER PROBLEME *
775 REM *
776 REM *****
780 CLS #1
790 PRINT #1, "VOTRE MISSION:":PRINT #1
800 PRINT #1, "VEILLER AU BON FONC-"
810 PRINT #1, "TIONNEMENT DE CETTE"
820 PRINT #1, "FERME EXPERIMENTALE.":PRINT #1
830 PRINT #1, "SONT EMBARQUES A BORD:"
840 PRINT #1, "- DES PLANTS DE TOMATE"
850 PRINT #1, "- DE COURGE"
860 PRINT #1, "- DE SALADE."
870 IF INKEY$="" THEN GOTO 870
880 CLS #1
890 PRINT #1, "TOUTES CES CULTURES "
900 PRINT #1, "OCCUPENT DES CAGETTES"
910 PRINT #1, "DANS LES JONCTIONS"
920 PRINT #1, "DE LIAISON."
930 LOCATE 9,13:PRINT "° ° ° ° °":LOCATE 32,13:P
RINT "S S S S S"
940 FOR I=0 TO 11
950 LET ID=I:GOSUB 2280
960 SOUND 1,100,25,15,0,0,0
970 SOUND 2,102,25,15,0,0,0
980 IF SQ(I)>4 THEN GOTO 980
990 NEXT I
1000 PRINT #1
1010 PRINT #1, "DEUX ASCENSEURS"
1020 PRINT #1, "SONT A VOTRE DISPO-"
1030 PRINT #1, "SITION POUR QUE VOUS"
1040 PRINT #1, "PUISSIEZ Y ACCEDER."
1050 PRINT #1
1060 PRINT #1, "LORS DE VOTRE ARRIVEE"
1070 PRINT #1, "LES PLANTS, ENCORE PETITS,"
1080 PRINT #1, "NE SONT PAS RANGES DANS"
1090 PRINT #1, "UN ORDRE BIEN PRECIS."
1100 PRINT #1
1110 PRINT #1, "VOTRE PREMIERE MISSION"
1120 PRINT #1, "EST DONC DE RANGER"
1130 PRINT #1, "LE JARDIN."
1140 SOUND 1,100,25,15,0,0,0
1150 SOUND 2,102,25,15,0,0,0
1160 IF INKEY$="" THEN GOTO 1160
1170 CLS #1:LET JARD=0
1180 PRINT #1, "POUR EVITER QU'ELLES"
1190 PRINT #1, "SF DECROCHENT LES COURGES"
1200 PRINT #1, "NE DOIVENT PAS ETRE "
1210 PRINT #1, "SOUMISES A UNE PESENTEUR"
1220 PRINT #1, "SUPPERIEURE (G) DE 3,5"
1230 PRINT #1
1240 PRINT #1, "JUSQU'A QUELLE DISTANCE"
1250 PRINT #1, "DU SAS D'ACCEUIL LES"
1260 PRINT #1, "PLACEZ VOUS ?"
1262 REM *****
1263 REM *
1264 REM * PRISE EN COMPTE DE LA REPONSE *
1265 REM *
1266 REM *****
1270 INPUT #1, "METRES":DM:IF DM>16 THEN PRINT #1, "
VOUS ETES HORS JONCTION!":GOTO 1270
1272 REM *****
1273 REM *
1274 REM * CALCUL DE LA VALEUR DE G. *
1275 REM *
1276 REM *****
1280 LET G=((((DM+6)*2*PI)/10.99)^2)/(DM+6)
1290 PRINT #1, "ICI G=";G
1292 REM *****
1293 REM *
1294 REM * COMMENTAIRES. *
1295 REM *
1296 REM *****
1300 IF G<=3.5 THEN GOTO 1360
1310 PRINT #1:PRINT #1, "VOS COURGES SE DECROCHENT"
1320 PRINT #1, "ENTRAINANT AVEC ELLES LE"
1330 PRINT #1, "RESTANT DU POTAGER."
1340 PRINT #1, "TOUTE VOTRE RECOLTE"
1350 PRINT #1, "S'ECRASE DANS LA COURSIVE.":GOTO 14
30
1360 IF G>3.3 THEN GOTO 1410
1370 PRINT #1:PRINT #1, "VOUS AURIEZ PU PLANTER"
1380 PRINT #1:PRINT #1, "PLUS DE COURGES;"
1390 PRINT #1:PRINT #1, "VOUS PERDEZ EN RENDEMENT!"
1400 LET JARD=INT(DM*2):PRINT #1, "RENDEMENT: ";10*J
ARD; "%":GOTO 1430
1410 PRINT #1:PRINT #1, "BRAVO; PARFAIT!"
1420 LET JARD=10
1430 SOUND 1,100,200,15,0,0,0
1440 SOUND 2,102,200,15,0,0,0
1450 IF INKEY$="" THEN GOTO 1450
1452 REM *****
1453 REM *
1454 REM * DEUXIEME PROBLEME *
1455 REM *
1456 REM *****
1460 CLS #1:LET POU=0
1470 PRINT #1, "UNE COUVEUSE REMPLIE"
1480 PRINT #1, "D'OEUF EST DANS LE SAS."
1490 FOR E=0 TO 3
1500 LET IK=E:LET D=40:GOSUB 2180
1510 NEXT E
1520 LOCATE 24,12:PRINT "####"
1530 LOCATE 24,13:PRINT "####"
1540 LOCATE 24,14:PRINT "####"

```

(ligne 1270) puis le calcul de la valeur de G (ligne 1280). Notons, à propos de cette ligne, que le symbole "6" correspond en fait à la flèche vers le haut placée sur la même touche que Livre Sterling; il signifie puissance. Ensuite, en fonction de la réponse fournie, la valeur de G sera suivie de commentaires sur l'état du jardin potager (des lignes 1300 à 1450).

Les données concernant le second problème sont affichées par les lignes de 1460 à 1660. Puis deux questions vous seront posées: le choix du parcours (des lignes 1670 à 1740) et la détermination de la vitesse de déplacement (des lignes

1750 à 1820). Ici encore, en fonction des réponses fournies et après calcul de G (ligne 1830), des commentaires seront présentés par les lignes de 1840 à 2060. Pour terminer, un récapitulatif global sur le déroulement de la partie sera fourni par les lignes de 2070 à 2150 et le jeu rebouclera sur la frappe d'une touche.

Restent les deux sous-routines. La première (des lignes 2180 à 2260) se charge de tracer les cercles composant le schéma de notre base et la seconde (des lignes 2280 à 2320) les 4 droites représentant les tubes de jonction.

Hormis le problème des symbo-

les "ˆ" des lignes 1280 et 1830 ainsi que ceux de la ligne 930 ("o" et "s") qui doivent respectivement être remplacés par crochet ouvert et crochet fermé — les deux touches situées à côté de la touche bleue ENTER), la frappe de ce programme ne doit pas poser de problème particulier si ce n'est sa longueur.

Après avoir demandé RUN, une première page d'accueil sera affichée, afin de combler le temps de calcul des coordonnées des tracés de notre base, il suffira ensuite de suivre les instructions présentées sur l'écran pour jouer.

Henri-Pierre Penel

```

1550 PRINT #1:PRINT #1,"LES POUSSINS COMMENCENT"
1560 PRINT #1,"A ECLORE...":PRINT #1
1570 PRINT #1,"IL FAUT LES ATTRAPER ET"
1580 PRINT #1,"LES CHARGER DANS L'UN OU"
1590 PRINT #1,"L'AUTRE DES TRIPORTEURS"
1600 PRINT #1,"EQUIPES DE PANNIERS"
1610 PRINT #1,"PLACES EN 1 ET 2 ET LES"
1620 PRINT #1,"CONDUIRE A LA PRAIRIE"
1630 PRINT #1,"ARTIFICIELLE EN 3."
1640 LOCATE 2,13:PRINT "1"
1650 LOCATE 49,13:PRINT "2"
1660 LOCATE 26,23:PRINT "3"
1662 REM *****
1663 REM *
1664 REM *          CHOIX DU PARCOURS          *
1665 REM *
1666 REM *****
1670 PRINT #1
1680 PRINT #1,"CHOISISSEZ VOUS LE"
1690 PRINT #1,"TRIPORTEUR 1 OU LE 2 ?"
1700 PRINT #1:PRINT #1,"TAPEZ VOTRE CHOIX."
1710 SOUND 1,100,200,15,0,0,0
1720 SOUND 2,102,200,15,0,0,0
1730 LET K$=INKEY$:IF K$<>"1" AND K$<>"2" THEN GOT
O 1730
1740 LET T$=K$
1742 REM *****
1743 REM *
1744 REM *          CHOIX DE LA VITESSE          *
1745 REM *
1746 REM *****
1750 CLS #1:PRINT #1,"LES TRIPORTEURS PEUVENT"
1760 PRINT #1,"ROULER A UNE VITESSE"
1770 PRINT #1,"COMPRISE ENTRE 0 ET"
1780 PRINT #1,"35 KM/H; A QUELLE"
1790 PRINT #1,"VITESSE ROULEZ-VOUS?"
1800 INPUT #1,"KM/H":VTR:IF VTR>35 THEN GOTO 1800
1810 LET VTR=VTR/3.6
1820 IF T$="2" THEN GOTO 1890
1822 REM *****
1823 REM *
1824 REM *          CALCUL DE G          *
1825 REM *
1826 REM *****
1830 LET G=(17.15-VTR)^2/30
1832 REM *****
1833 REM *
1834 REM *          COMMENTAIRES          *
1835 REM *
1836 REM *****
1840 PRINT #1,"LORS DE VOTRE DEPLACEMENT"
1850 PRINT #1,"G A ETE DE":G
1860 IF G>3 AND G<=15 THEN PRINT #1,"TOUT S'EST BI
EN PASSE":PRINT #1,"MAIS CE N'EST PAS RAPIDE":LET
POUS=10:GOTO 1900
1870 IF G<=3 THEN PRINT #1,"G EST TROP FAIBLE":PRI
NT #1,"DES POUSSINS S'ECHAPENT":GOTO 1900
1880 IF G>15 THEN PRINT #1,"G EST TROP FORT DES":P
RINT #1,"POUSSINS MEURENT ECRASES":GOTO 1900
1890 LET VTR=-VTR:GOTO 1830
1900 IF POUS=10 THEN GOTO 2040
1910 IF G<5 THEN GOTO 1950
1920 LET POUS=10-INT(G-11)
1930 IF POUS<0 THEN LET POUS=0
1940 PRINT #1,"RENDEMENT:":POUS*10:"%":GOTO 2040
1950 LET POUS=INT(3*G)
1960 PRINT #1,"RENDEMENT:":POUS*10:"%"
1970 PRINT #1,"DE PLUS LES POUSSINS"
1980 PRINT #1,"ECHAPPES ONT ATTAQUES"
1990 PRINT #1,"VOS CULTURES.."
2000 LET JARD=JARD-(10-POUS)
2010 IF JARD<0 THEN LET JARD=0
2020 PRINT #1:PRINT #1,"POUR CELLES CI LE NOUVEAU"
2030 PRINT #1:PRINT #1,"RENDEMENT EST DE:":JARD*10
:"%"
2040 SOUND 1,100,200,15,0,0,0
2050 SOUND 2,102,200,15,0,0,0
2060 IF INKEY$=" " THEN GOTO 2060
2062 REM *****
2063 REM *
2064 REM *          RECAPITULATIF DE LA PARTIE          *
2065 REM *
2066 REM *****
2070 CLS
2080 IF JARD=10 AND POUS=10 THEN PRINT "BRAVO":GOT
O 2140
2090 PRINT "RECAPITULATIF:"
2100 PRINT "JARDIN POTAGER RENDEMENT:":JARD*10:"%"
2110 PRINT "ELEVAGE DE POUSSINS:":POUS*10:"%"
2120 PRINT:PRINT "VOUS AVEZ ENCORE DES PROGRES A F
AIRE."
2130 PRINT:PRINT "RELISEZ BIEN LES CARACTERISTIQUE
S DE VOTRE BASE."
2140 LOCATE 10,10:PRINT "POUR UN AUTRE ESSAI TAPEZ
UNE TOUCHE S.V.P."
2150 IF INKEY$=" " THEN GOTO 2150
2152 REM *****
2153 REM *
2154 REM *          REBOUCLAGE DU JEU          *
2155 REM *
2156 REM *****
2160 GOTO 290
2170 REM *****
2173 REM *
2174 REM * SOUS ROUTINE DE TRACÉ DES CERCLES *
2175 REM *
2176 REM *****
2180 PLOT 200,200+(D*0.9)
2190 PLOT 200,200+(D*0.9)
2200 FOR I=0 TO 360 STEP 10
2210 LET X=200+(D*SIN(I)):LET Y=200+(D*0.9*COS(I))
2220 DRAW X,Y,IK
2230 SOUND 1,(400-X)/10,2,15,0,0,0
2240 SOUND 2,(400-Y)/10,2,15,0,0,0
2250 NEXT I
2260 RETURN
2270 REM *****
2273 REM *
2274 REM * SOUS ROUTINE DE TRACE DES DROITES *
2275 REM *
2276 REM *****
2280 PLOT J(Z),K(Z):DRAW L(Z),M(Z),ID
2290 PLOT 400-J(Z),400-K(Z):DRAW 400-L(Z),400-M(Z)
,ID
2300 PLOT N(Z),O(Z):DRAW P(Z),Q(Z),ID
2310 PLOT 400-N(Z),400-O(Z):DRAW 400-P(Z),400-Q(Z)
,ID
2320 RETURN

```